

# SAS/ODS (Output Delivery System) - eine Einführung

Heribert Ramroth  
Universitätsklinikum Heidelberg  
Institut für Public Health  
Im Neuenheimer Feld 324  
69120 Heidelberg  
Heribert.Ramroth@uni-heidelberg.de

## Zusammenfassung

SAS-ODS, das Output Delivery System, ist für den SAS-Benutzer ein effizientes und unverzichtbares Werkzeug. SAS-ODS ermöglicht eine Aufbereitung bzw. Weiterverarbeitung des SAS Outputs und stellt außerdem leicht zugängliche Schnittstellen zu anderen Softwarepaketen (Textverarbeitung (rtf, pdf, etc.), Tabellenkalkulation (csv, etc.) zur Verfügung. Der Vortrag soll aufzeigen, wie man in der alltäglichen Programmierpraxis ODS effektiv einsetzen kann.

**Schlüsselwörter:** SAS ODS, ODS, ODS RTF

## 1 Motivation

SAS ODS ermöglicht für alle Prozeduren eine einheitliche Vorgehensweise Ausgabedateien zu erzeugen. Wer dieses einfache Prinzip erkannt hat, kann die Ausgabe einer beliebigen Prozedur oft mit recht einfachen programmtechnischen Mitteln nachbearbeiten. Da die übersichtliche Darstellung von Studienergebnissen sowohl in der alltäglichen Praxis, als auch zur Erstellung von Publikationen sinnvoll und wünschenswert ist, konzentriert sich dieser Artikel hauptsächlich auf die Weiterverarbeitung von Prozedurergebnissen und deren Ausgabe im rtf-Format zur weiteren Textverarbeitung. Insbesondere gibt dieser Beitrag die Erfahrung des Autors wieder und hat nicht den Anspruch, einen vollständigen Überblick über die Möglichkeiten von SAS ODS zu geben. Obwohl dieser Beitrag nur eine Einführung sein kann, werden durch die Anwendung der hier vorgestellten Techniken einfache Mittel gezeigt, Programmstruktur und -ausgabe effektiv zu verbessern.

### 1.1 Überblick

- ODS-Ausgabeziele (Destinations)
- Stöbern in der ODS Ausgabe
- Arbeiten mit und Bearbeiten von ODS Ausgabetablen
- Selektieren der ODS Ausgabe
- Beispiele

Als Beispiel dient hier i.Allg. ein Datensatz mit epidemiologischem Hintergrund: Eine gematchte Fall-Kontrollstudie zu mütterlichen Risikofaktoren im Zusammenhang mit

untergewichtigen Geburten, der aus der SAS Hilfe kopiert werden kann (PROC PHREG Beispiel siehe Literatur). Hosmer & Lemeshow (1989), *Applied Logistic Regression*, John Wiley & Sons. Als Beispielprozeduren werden PROC PRINT, PROC FREQ, PROC PHREG und PROC CONTENTS verwendet.

## 1.2 Warum ODS?

Vorteile von ODS im Überblick:

- Ergebnisse lesbarer machen (insbesondere für “Nicht-SAS-Benutzer”)
- Hilfreich bei der Erstellung von Präsentationen / Artikeln
- Erzeugen von .rtf / .html / .pdf / ... - Dateien
- Erzeugen von Ausgabe Datensätzen
- ODS ist schon in seinen Grundzügen ein einfaches aber effektives Werkzeug!

## 1.3 ODS-Ausgabeziele

Der vorliegende Beitrag beschränkt sich hauptsächlich auf folgende ODS-Ausgabeziele (engl. destinations):

- ODS LISTING
- ODS OUTPUT
- ODS RTF

Da es sich bei dem vorliegenden Beitrag um eine Einführung in ODS handelt, werden weitere ODS-Ausgabeziele wie zum Beispiel ODS HTML, ODS PDF, ODS PRINTER, ODS LATEX, etc. nur auszugsweise angedeutet bzw. nicht behandelt.

ODS-Ausgabeziele müssen in der Regel vor Gebrauch geöffnet und nach Gebrauch geschlossen werden.

```
ods <destination> <options>;  
... SAS statements ...  
ods <destination> close;
```

### ODS-Listing

Das Ausgabeziel ODS Listing benutzt man, auch wenn man bisher noch nicht bewusst von SAS/ODS gehört hat und bezeichnet die Ausgabe der SAS-Ergebnisse sowohl im “RESULTS” Fenster (linkes Feld), als auch “OUTPUT” Fenster (rechtes Feld).

```
ods listing;  
... SAS statements ...  
ods listing close;
```

Diese Ergebnisse sind temporär, bzw. lassen sich bei Bedarf permanent als “.lst”-Datei abspeichern. Das Ausgabeziel Listing ist normalerweise automatisch geöffnet und kann, muss aber nicht notwendigerweise geschlossen werden.

## ODS/RTF

ODS RTF ist die Schnittstelle zur Textverarbeitung. Mit einfachem Code lässt sich die Ausgabe der SAS Ergebnisse in ein rtf-Dokument umleiten, welches dann in einem Textverarbeitungsprogramm (MS Word, Open Office) weiter bearbeitet werden kann.

```
ODS rtf file=filename <options>;
... SAS Statements ...;
ODS rtf close;
```

Ein einfaches Beispiel:

```
ODS rtf file="test.rtf";
proc print data=LBW;
run;
ODS rtf close;
```

Die Datei test.rtf wird im aktuell eingestellten Arbeitspfad abgelegt. Soll die Datei unter einem anderen Pfad abgelegt werden, lässt sich dieser vorab definieren. Der Code beim Öffnen des RTF-Ausgabeziels lautet dann folgendermaßen:

```
ods rtf file="C:\MyProject\test.rtf";
```

oder per filename Angabe vorab definiert:

```
filename RtfName "C:\MyProject\test.rtf";
ods rtf file= RtfName;
proc print data=LBW;
run;
ods rtf close;
```

Im Results-Fenster wird nun zusätzlich zu der bekannten Ausgabe noch das mit obigem Code erzeugte rtf-Dokument angezeigt. Außerdem lassen sich die nachfolgenden Optionen in den SAS Grundeinstellungen unter „Tools → Options → Preferences“ in der Registerkarte „Results“ setzen:

- „view results as they were generated“ und „Preferred web browser“:  
Das rtf-Dokument wird automatisch von Ihrem Textverarbeitungsprogramm geöffnet.
- „view results as they were generated“ und „Internal browser“:  
Das rtf-Dokument wird in einem „Results Viewer“-Fenster (neben EDITOR-, LOG- und OUTPUT-Fenster) angezeigt.
- Ohne Option „view results as they were generated“:  
Das rtf-Dokument wird erstellt, aber nicht automatisch angezeigt.

Sinnvolle ODS RTF Optionen sind die Folgenden:

- BODYTITLE  
Titel erscheinen im Rumpf des Textdokumentes anstatt in der Kopfzeile.

- **STARTPAGE=NO**  
Verhindert die standardmäßigen Seitenumbrüche, die bei der SAS Ausgabe die Regel sind. Somit lässt sich eine Menge Papier einsparen.
- **Style Optionen:**  
ODS RTF hält verschiedene Ausgabestile in einer Bibliothek bereit (styles library). Ohne explizite Stilangabe wird ein vorgegebener Stil verwendet, der in den Tabellen die erste Zeile und die linke Spalte grau schattiert hinterlegt, sowie die Tabelle mit Gittermuster ausgibt. Verhindert wird dies durch explizite Wahl eines Stils, zum Beispiel der Stilangabe „minimal“ oder besser noch „journal“, der direkt schon für die wissenschaftliche Ausgabe geeignet ist (style=styles.journal). Die Beispiele sehen wie folgt aus:

a) keine Stil-Angabe:

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0

b) Stil „minimal“:

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0

c) Stil „journal“:

---

<i>Obs</i>	<i>Name</i>	<i>Sex</i>	<i>Age</i>	<i>Height</i>	<i>Weight</i>
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0

---

Zusammengefasst schreiben sich obige RTF-Optionen dann wie folgt:

```
ods rtf file=RtfFile bodytitle startpage=no
      style=styles.journal;
proc print data = LBW;
run;
ods rtf close;
```

Doch, wer will sich schon diesen überaus nützlichen, doch sehr langen Text schon merken? Am besten man übergibt diesen an eine globale Makrovariable, um sie später an einer beliebigen Programmstelle aufzurufen (siehe: Proceedings KSFE 2009, Tipps & Tricks).

```
Filename RtfDummy "C:\MyProject\test.rtf";
%LET OdsRtfOpen = ODS rtf file=RtfDummy bodytitle
                    startpage=no style=styles.journal;
&OdsRtfOpen.;
    proc print data = LBW;
    run;
ods rtf close;
```

Unter SAS 9.2 gibt es weitere sinnvolle ODS RTF Optionen:

- **BODYTITLE\_AUX**

Die Titel im Rumpf des Textdokumentes lassen sich in Textkästen positionieren:

```
ods rtf file="test.rtf" BODYTITLE_AUX;
title j=1 "left" j=c "center" j=r "right";
```

- **ODS TEXT=** "Ich schreibe hier jetzt mal was zwischenrein..."  
Ermöglicht, die Ausgabe zu Kommentieren bzw. Ergänzen.

- **ODS RTF PAGE;**  
Fügt einen neuen Seitenumbruch ein.

- **CONTENTS TOC\_DATA**

Ermöglicht das Einfügen eines Inhaltsverzeichnisses.

```
ods rtf file="test.rtf" contents toc_data;
proc contents data = ...;
proc freq data = ...;
proc phreg data = ...;
run;
ods rtf close;
```

Als erste Zeile erscheint im RTF-Dokument ein Textblock mit der Beschriftung: „Table of Contents“. Dieses Inhaltsverzeichnis erscheint jedoch nicht automatisch. Als nächsten Schritt muss man nun mit einem rechten Mausklick direkt in die linke Ecke der nächsten Zeile klicken und im erscheinenden Menü „Felder aktualisieren“ auswählen.

- **OPTIONS ORIENTATION=LANDSCAPE** (vs. PORTRAIT – generelle SAS Option !)

Ermöglicht das Ändern der Seitenausrichtung bei geöffnetem RTF-Dokument. Die Zwischenzeile „ODS RTF;“ bewirkt die Umsetzung des Codes.

```
options orientation=portrait;
ods rtf file=RtfDummy;
  title 'Page Orientation: Portrait';
  proc print data= ...;
  run;
  title 'Page Orientation: Landscape';
  options orientation=landscape;
  ods rtf;
  proc print data= ...;
  run;
ods rtf close;
```

- ODS IMAGE\_DPI=  
Ermöglicht DPI Einstellungen von Grafiken in RTF Dokumenten.

## ODS/HTML

Ein HTML-Dokument lässt sich genauso leicht erzeugen, wie ein einfaches RTF-Dokument (wie oben: optional filename statement). Auch hier wird ein neues Dokument im Results-Fenster angelegt.

```
ods html body="C:\MyProject\test.html";
  proc print data= ...;
  run;
ods html close;
```

Und noch eine vom Ausgabeziel unabhängige ODS Option:

- ODS NOPTITLE (vs. ODS PTITLE)  
Unterdrückt die Titel der Prozeduren bzw. stellt sie wieder her.

### 1.4 Stöbern in der ODS-Ausgabe

Mit Einführung von SAS ODS hat jegliche Ausgabe, die im Ausgabe-Fenster erscheint, einen Namen, mit dem diese Ausgabe für weitere Zwecke bearbeitet werden kann. Somit sind die alten OUT-Optionen, die teilweise prozedurspezifisch vorhanden waren, auf einheitliche Weise ersetzbar. Die Aufgabe besteht lediglich darin herauszufinden, wie die jeweiligen Namen lauten.

- ODS TRACE ON;  
Im LOG-Fenster erscheint bei Aufruf der Prozedur ein Informationsteil, der Name, Bezeichnung (Label) und weitere Informationen des Ausgabeteils angibt.

- ODS TRACE ON / LISTING;  
Bewirkt die gleiche Aktion wie oben, nur dass die Information nicht im LOG-Fenster erscheint, sondern sinnvollerweise vor dem jeweiligen Ausgabeteil im OUTPUT-Fenster. Dies wird am Beispiel der PROC PHREG demonstriert.

```
ods trace on / listing;
proc phreg data = LBW;
model Time * Low(0) = LWT Smoke HT UI/ties=discrete rl;
strata Age;
run;
ods trace off;
```

Die Ausgabe dieser Prozedur hat 7 Teile mit verschiedenen Namen, unter denen sie für die weitere Verwendung ansteuerbar sind. Man beachte, dass es sich hiermit noch nicht um temporäre Dateien handelt. Diese lassen sich erst mit der gleich aufgeführten ODS Anweisung: ODS OUTPUT realisieren.

Angenommen man interessiert sich für die Hazard Ratios/Odds Ratios mit den zugehörigen 95%-Konfidenzintervallen, dann findet man Name und Bezeichnung (Label) dieses Ergebnisteils im OUTPUT-Fenster im zugehörigen Vorlaufertext:

Name: ParameterEstimates

Bezeichnung: „Maximum Likelihood Estimates for Model Parameters“.

Die Erfahrung zeigt, dass Namen i. Allg. kürzer und handlicher sind.

- ODS TRACE OFF;  
Wenn der Schalter einmal auf ON gestellt wurde, werden diese Informationen solange angezeigt, bis man ihn wieder auf OFF stellt.

Da die Namen generell pro Prozedur gelten, d.h. bei jedem Aufruf ist dies sinnvoll, da man sich ziemlich schnell an die Namen gewöhnt. Eine dem Autor bekannte Ausnahme stellt hier PROC FREQ dar:

a) `proc freq data=LBW; tables LOW; run;`

b) `proc freq data=LBW; tables LOW * Smoke; run;`

In Fall a) werden die Häufigkeiten in einem Ergebnisteil namens OneWayFreqs zur Verfügung gestellt, in Fall b) mit Namen CrossTabFreqs.

- ODS OUTPUT  
Über diese Anweisung lässt sich nun das von ODS zur Verfügung gestellte Ergebnis in einer Datei abspeichern, deren Namen man selbst vergibt. In einer Anweisung können mehrere Ausgabeteile benannt werden. Allgemein:  
`ods output <Name>=Eigener_Name;`  
oder:  
`ods output <"Label">=Eigener_Name;`

Beispiel:

```
ods output ParameterEstimates=ParmEst;  
proc phreg data = LBW;  
model Time * Low(0)= LWT Smoke HT UI/ties=discrete rl;  
strata Age;  
run;
```

Die Ergebnisse der Prozedur werden weiterhin alle im OUTPUT-Fenster angezeigt. Nur der Ergebnisteil mit den Odds Ratios (p-Werte, Konfidenzgrenzen, etc.) werden in einer temporären Datei namens ParmEst abgelegt.

Der Sinn einer solchen Aktion könnte zum Beispiel darin liegen, die Konfidenzintervalle nun nicht per Hand in eine Ergebnistabelle zu übertragen, sondern per Datenschnitt zu erzeugen. Mit der Funktion CATS werden anführende und schließende Klammern, gerundete Konfidenzgrenzen, sowie ein zwischengeschaltetes Komma zusammengefügt:

```
data ParmEst1;  
set ParmEst;  
OR=PUT(HazardRatio, 5.1);  
CI=CATS ("(", PUT(HRLowerCL, 5.1), ",", PUT(HRUpperCL, 5.1), ")");  
run;
```

Mit einem weiteren Leerzeichen zwischen den Konfidenzgrenzen sieht das Konfidenzintervall jedoch sicherlich schöner aus:

```
CI= CATS ("(", PUT(HRLowerCL, 5.1)) ||", " ||  
          CATS(PUT(HRUpperCL, 5.1), ")" );
```

Der hier abgebildete Code lässt sich zur Erzeugung schönerer Intervalle noch verbessern (nachzulesen unter: Ramroth, Proceedings KSFE 2008).

Die der Prozedur entweder vorangehende, zumindest vor dem nächsten RUN stehende ODS OUTPUT Anweisung gilt nur für den nächsten Prozeduraufruf. Mit der zusätzlichen Option (PERSIST) kann dies jedoch auch dauerhaft eingestellt werden. Dabei sollte man beachten, dass man bei mehreren Prozedurabläufen auch den Überblick über die Ausgabedateien behält und evtl. alte Prozedurausgaben löscht.

- ODS SELECT
- ODS EXCLUDE

Allgemein: ODS SELECT <ODS-destination> output-object(s);  
            ODS EXCLUDE <ODS-destination> output-object(s);

Mit dieser Anweisung können bestimmte Ergebnisse ausgewählt oder ausgeschlossen werden. Dies gilt sowohl allgemein, als auch nur für bestimmte Ausga-



beziele (destinations). Wie bei ODS OUTPUT gilt diese Anweisung nur für den nächsten Prozeduraufruf und muss bei Bedarf mit der Option (PERSIST) dauerhaft eingestellt werden.

```
ods select ParameterEstimates;
proc phreg data =LBW;
    model Time*Low(0)= LWT Smoke HT UI / ties =discrete;
    strata Age;
run;
```

- ODS SELECT ALL;  
Stellt die ursprüngliche Auswahl wieder her.
- ODS SHOW;  
Sollte man den Überblick verloren haben, zeigt diese ODS Anweisung Details der aktuellen Auswahl im LOG Fenster.

## ZUSAMMENFASSUNG

Nicht nur für den Wissenschaftler zeigen schon die wenigen hier vorgestellten ODS Optionen hinreichende Möglichkeiten auf, die Ergebnisausgabe effektiver zu gestalten.

Ein wichtiger Hinweis: Bei Umstellung von SAS 9.1 zu SAS 9.2 wurden innerhalb der Tabelle ParameterEstimates von PROC PHREG Variablennamen geändert (alt: „Variable“ → neu „Parameter“).

Es ist sicher möglich, dass es weitere, ODS betreffende Änderungen gibt.

## Literatur

- [1] Ramroth H (2008): Publikationsfertige Kombination von Häufigkeiten und Risiko-Kennwerten aus Ergebnissen von klinisch-epidemiologischen Studien; Proceedings der 12. KSFE; Aachen.
- [2] Tipps und Tricks (KSFE 2009): Proceedings der 13. KSFE; Halle/Saale.
- [3] Ramroth H (2009): Makros zur Kombination von deskriptiven und analytischen Ergebnissen aus klinisch-epidemiologischen Studien; Proceedings der 13. KSFE; Halle/Saale.
- [4] Hosmer DWJ & Lemeshow S (1989) *Applied Logistic Regression*, New York: John Wiley & Sons, Inc.
- [5] PROC PHREG-Example : Conditional Logistic Regression for m:n Matching.

